Problem:
Score : 200 points

Problem Statement

Takahashi went to an all-you-can-eat buffet with N kinds of dishes and ate all of them (Dish 1, Dish 2, \ldots, Dish N) once.

The i-th dish (1 \leq i \leq N) he ate was Dish A_i.

When he eats Dish i (1 \leq i \leq N), he gains B_i satisfaction points.

Additionally, when he eats Dish i+1 just after eating Dish i (1 \leq i \leq N - 1), he gains C_i more satisfaction points.

Find the sum of the satisfaction points he gained.

Constraints

All values in input are integers.

2 \leq N \leq 20

1 \leq A_i \leq N

A_1, A_2, ..., A_N are all different.

1 \leq B_i \leq 50

1 \leq C_i \leq 50

Input

Input is given from Standard Input in the following format:

N
A_1 A_2 ... A_N
B_1 B_2 ... B_N
C_1 C_2 ... C_{N-1}

Output

Print the sum of the satisfaction points Takahashi gained, as an integer.

Sample Input 1

3
3 1 2
2 5 4
3 6

Sample Output 1

14

Takahashi gained 14 satisfaction points in total, as follows:

First, he ate Dish 3 and gained 4 satisfaction points.

Next, he ate Dish 1 and gained 2 satisfaction points.

Lastly, he ate Dish 2 and gained 5 + 3 = 8 satisfaction points.

Sample Input 2

4
2 3 4 1
13 5 8 24
45 9 15

Sample Output 2

74

Sample Input 3

2
1 2
50 50
50

Sample Output 3

150

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

module ABC140

  ! <module>s to import
  use, intrinsic :: iso_fortran_env

  ! require all variables to be explicitly declared
  implicit none

  ! accessibility of <subroutine>s and <function>s in this <module>
  public :: task_B

  ! contained <subroutine>s and <function>s are below
  contains

  subroutine task_B

    ! variables for this <subroutine>
    integer(INT32) :: num_dishes
    integer(INT32) :: sum_satisfaction

    ! arrays for this <subroutine>
    integer(INT32), dimension(:), allocatable :: list_turn
    integer(INT32), dimension(:), allocatable :: list_satisfaction
    integer(INT32), dimension(:), allocatable :: list_satisfaction_additional

    ! support variables for this <subroutine>
    integer :: itr

    ! STEP.01
    ! read out the given data
    ! (the number of the dishes)
    read(unit=INPUT_UNIT, fmt=*) num_dishes

    ! STEP.02
    ! allocate the arrays to store the given data
    allocate( list_turn(1:num_dishes) )
    allocate( list_satisfaction(1:num_dishes) )
    allocate( list_satisfaction_additional(1:num_dishes - 1) )

    ! STEP.03
    ! read out the given data
    read(unit=INPUT_UNIT, fmt=*) list_turn(:)
    read(unit=INPUT_UNIT, fmt=*) list_satisfaction(:)
    read(unit=INPUT_UNIT, fmt=*) list_satisfaction_additional(:)

    ! STEP.04
    ! calculate the sum of the satisfaction
    sum_satisfaction = list_satisfaction( list_turn(1) )

    do itr = 2, num_dishes, 1

      sum_satisfaction = sum_satisfaction + list_satisfaction( list_turn(itr) )

      if ( list_turn(itr - 1) + 1_INT32 .eq. list_turn(itr) ) then
        sum_satisfaction = sum_satisfaction + list_satisfaction_additional( list_turn(itr - 1) )
      end if

    end do

    ! STEP.05
    ! output the answer of this task
    write(unit=OUTPUT_UNIT, fmt='(I0)', advance='yes') sum_satisfaction

    ! STEP.06
    ! deallocate the arrays to store the given data
    deallocate( list_turn )
    deallocate( list_satisfaction )
    deallocate( list_satisfaction_additional )

    ! STEP.END
    return

  end subroutine task_B

end module ABC140


program main

  ! <module>s to import
  use, non_intrinsic :: ABC140

  ! require all variables to be explicitly declared
  implicit none

  call task_B

end program main